cap cd "C:\Users\ejm5\Dropbox\PAPI conjoint\Replication"
cap cd "/Users/paulschuler/Dropbox/projects/PAPI conjoint/Replication"
use PAPI_Conjoint_3years.dta, clear
set more off
 
tab year
lab var party "Party Member"

mvdecode d101* d305* vote, mv(888=.b\999=.c)


#delimit;
label values a011 perform;
label define perform 0 "Very Bad" 1 "Bad" 2 "Neutral" 3"Good" 4"Very Good";

/*Demonstration that Survey Weights are More Restrictive Than Clustering at Individual Level*/

/*Appendix D*/
#delimit;
svy: reg vote party if year==2016;
outreg2 using "Tables\AppendixD", tdec(3) bdec(3) e(N_clust N_strata N_psu N_poststrata)  replace;
reg vote party if year==2016, cluster(respondent_id);
outreg2 using "Tables\AppendixD", tdec(3) bdec(3) e(N_clust N_strata N_psu N_poststrata) ;
svy: reg vote party if year==2017;
outreg2 using "Tables\AppendixD", tdec(3) bdec(3) e(N_clust N_strata N_psu N_poststrata)  ;
reg vote party if year==2017, cluster(respondent_id);
outreg2 using "Tables\AppendixD", tdec(3) bdec(3) e(N_clust N_strata N_psu N_poststrata) ;
svy: reg vote party if year==2018;
outreg2 using "Tables\AppendixD", tdec(3) bdec(3) e(N_clust N_strata N_psu N_poststrata) ;
reg vote party if year==2018, cluster(respondent_id);
outreg2 using "Tables\AppendixD", tdec(3) bdec(3) e(N_clust N_strata N_psu N_poststrata)  excel;



/*Figure 2: Full Analysis of All Exposed to Experiment and CI Plot*/

/*2016*/
#delimit;
set more off;
svy: reg  vote  i.trade i.poverty i.environment  i.VFF   i.age_39 i.male i.home  i.gov_nominee i.party if year==2016;
margins if year==2016, dydx( i.trade i.poverty i.environment i.VFF   i.age_39 i.male i.home  i.gov_nominee i.party );
marginsplot,  horizontal plotopts(connect(none)) xline(0, lpattern(dash) lwidth(thick) lcolor(maroon)) title("")
yline(9, lwidth(vvvthick) lcolor(ltblue)) scheme(s1mono) recastci(rcap) ciopts(lcolor(navy) lwidth(thick) mcolor(navy) msymbol(i) msize(vtiny))
ylabel( 1 "Trade" 2 "Poverty" 3 "Environment" 4 "Mass Organization"  5 `""Younger""Candidate""' 6 "Male"  
7 `""Home""Province""' 8 `""Officially""Nominated""' 9 `""Party""Member""' , labsize(vsmall)) 
xtitle("AMCE on Vote Choice", size(small) margin(medium))
xlab(-.2(.05).25, labsize(vsmall)) ytitle("") title("2016 PAPI Survey (n=12,916)", size(medium));
graph save "Figures\Figure2_CIs_2016.gph", replace;


/*2017*/
#delimit;
set more off;
svy: reg vote  i.trade i.poverty i.environment i.VFF i.age_39 i.male i.home  i.gov_nominee i.party if year==2017;
margins if year==2017, dydx( i.trade i.poverty i.environment i.VFF i.age_39 i.male i.home i.gov_nominee i.party );
marginsplot,  horizontal plotopts(connect(none)) xline(0, lpattern(dash) lwidth(thick) lcolor(maroon)) title("")
yline(9, lwidth(vvvthick) lcolor(ltblue)) scheme(s1mono) recastci(rcap) ciopts(lcolor(navy) lwidth(thick) mcolor(navy) msymbol(i) msize(tiny))
ylabel( 1 "Trade" 2 "Poverty" 3 "Environment" 4 "Mass Organization"  5 `""Younger""Candidate""' 6 "Male"  
7 `""Home""Province""' 8 `""Officially""Nominated""' 9 `""Party""Member""' , labsize(vsmall))  
xtitle("AMCE on Vote Choice", size(small) margin(medium))
xlab(-.2(.05).25, labsize(vsmall)) ytitle("") title("2017 PAPI Survey (n=11,736)", size(medium));
graph save "Figures\Figure2_CIs_2017_SHORT.gph", replace;


/*2018*/
#delimit;
set more off;
svy: reg vote i.trade i.poverty i.environment i.VFF i.age_39 i.male i.home  i.gov_nominee i.party if year==2018;
margins if year==2018, dydx( i.trade i.poverty i.environment i.VFF i.age_39 i.male i.home i.gov_nominee i.party );
marginsplot,  horizontal plotopts(connect(none)) xline(0, lpattern(dash) lwidth(thick) lcolor(maroon)) title("")
yline(9, lwidth(vvvthick) lcolor(ltblue)) scheme(s1mono) recastci(rcap) ciopts(lcolor(navy) lwidth(thick) mcolor(navy) msymbol(i) msize(tiny))
ylabel( 1 "Trade" 2 "Poverty" 3 "Environment" 4 "Mass Organization"  5 `""Younger""Candidate""' 6 "Male"  
7 `""Home""Province""' 8 `""Officially""Nominated""' 9 `""Party""Member""' , labsize(vsmall))  
xtitle("AMCE on Vote Choice", size(small) margin(medium))
xlab(-.2(.05).25, labsize(vsmall)) ytitle("") title("2018 PAPI Survey (n=10,620)", size(medium));
graph save "Figures\Figure2_CIs_2018_SHORT.gph", replace;





#delimit;
graph combine "Figures\Figure2_CIs_2016.gph" "Figures\Figure2_CIs_2017_short.gph" "Figures\Figure2_CIs_2018_short.gph", imargin(tiny) xcommon cols(3) scheme(s1mono) note("Based on Models 1 and 5 in Table 5; 95% CIs, Calculated with survey weights that address SE clustering at district, commune, and village levels", size(vsmall));
graph save "Figures\Figure2_All.gph", replace;
graph export "Figures\Figure2_All.pdf", replace;




/******************************************************************************************************************************/
/*Table 4*/

#delimit;
mean d101d5a, over(year);
mean d101d5b if d101d5==0 , over(year);
mean d305f, over(year);

svy: reg vote party non_party if year==2016, nocons;
svy: reg vote party non_party if year==2017, nocons;
svy: reg vote party non_party if year==2018, nocons;

#delimit;
gen nr_d101d5a=1 if d101d5a ==.b|d101d5a ==.c;
replace nr_d101d5a=1 if nr_d101d5a==. & d101d5==1;
replace nr_d101d5a=0 if d101d5a <=1;
by year, sort: sum  nr_d101d5a;

#delimit;
gen nr_d101d5b=1 if d101d5b ==.b|d101d5b ==.c;
replace nr_d101d5b=1 if nr_d101d5b==. & d101d5==0;
replace nr_d101d5b=0 if d101d5b <=1;
by year, sort: sum  nr_d101d5b;

#delimit;
gen nr_vote=1 if vote ==.b|vote ==.c;
replace nr_vote=0 if vote <=1;
replace nr_vote=1 if d102a==.b|d102a==.c;
replace nr_vote=1 if d102b==.b|d102b==.c;
by year, sort: sum nr_vote;

#delimit;
gen nr_d305f=1 if d305f ==.b|d305f ==.c;
replace nr_d305f=0 if d305f <=100;
by year, sort: sum  nr_d305f;


/*Table Appendix*/
#delimit;
sum nr*;
mvdecode d101d5a d101d5b d305f, mv(888=.\999=.);
#delimit;
sum d101d5a d101d5b d305f if year==2016;
sum d101d5a d101d5b d305f if year==2017;
sum d101d5a d101d5b d305f if year==2018;
reg vote party non_party if year==2016, nocons;
reg vote party non_party if year==2017, nocons;
reg vote party non_party if year==2018, nocons;

/******************************************************************************************************************************/


/*Table 5*/

/*2016*/

/*All*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx  replace;

/*Remove Preference Falsification*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016 & d101d5b==1;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;

/*Removing Performance Legitimacy*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


/*Same Policy and Economics Conditions are Neutral*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF trade poverty environment  if year==2016 &  d101d5b==1 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


/*2017*/

/*All*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;

/*Removing Preference Falsification*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017 & d101d5b==1;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


/*Removing Performance Legitimacy*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


#delimit;
set more off;
svy: reg vote  party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017  &  d101d5b==1 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;



/*2018 JOBS*/

/*All*/
#delimit;
set more off;
svy: reg vote_job party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy  if year==2018;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;

/*Removing Preference Falsification*/
#delimit;
set more off;
svy: reg vote_job party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy  if year==2018 & d101d5b==1;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


/*Removing Performance Legitimacy*/
#delimit;
set more off;
svy: reg vote_job party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy  if year==2018 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


#delimit;
set more off;
svy: reg vote_job party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy  if year==2018  &  d101d5b==1 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;

/*2018 EXPERIENCE*/

/*All*/
#delimit;
set more off;
svy: reg vote_exp party gov_nominee age_39 male home bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5  if year==2018;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;

/*Removing Preference Falsification*/
#delimit;
set more off;
svy: reg vote_exp party gov_nominee age_39 male home   bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5   if year==2018 & d101d5b==1;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


/*Removing Performance Legitimacy*/
#delimit;
set more off;
svy: reg vote_exp party gov_nominee age_39 male home   bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5  if year==2018 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx;


#delimit;
set more off;
svy: reg vote_exp party gov_nominee age_39 male home  bachelors  masters doctorate trade poverty environment no_policy exp_zero exp_y5 if year==2018  &  d101d5b==1 & a011==2;
mfx;
outreg2 using "Tables\Table5", tdec(3) bdec(3) e(Xmfx_y F N_strata N_psu N_poststrata) mfx excel;


/*******************************Different Versions of D101 to Address Preference Falsification***************************************/
/*All Versions of D101*/

/*2016*/

/*Voted for Non-Party Candidate*/

#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016 & d101d5a==1;
estimates store d101d5a_0_2016;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(F N_strata N_psu N_poststrata) excel;


#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016 & d101d5a==0;
estimates store d101d5a_1_2016;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all) ;

/*Would Vote for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016 & d101d5b==1;
estimates store d101d5b_0_2016;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);

/*Would Not for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF trade poverty environment if year==2016 & d101d5b==0;
estimates store d101d5b_1_2016;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);


/*2017*/
/*Voted for Non-Party Candidate*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017 & d101d5a==1;
estimates store d101d5a_0_2017;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);
/*Did Not Vote for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017 & d101d5a==0;
estimates store d101d5a_1_2017;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);
/*Would Vote for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017 & d101d5b==1;
estimates store d101d5b_0_2017;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);
/*Would Not for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy if year==2017 & d101d5b==0;
estimates store d101d5b_1_2017;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all) excel;


/*2018*/
/*Voted for Non-Party Candidate*/
#delimit;
set more off;
svy: reg vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5 if year==2018 & d101d5a==1;
estimates store d101d5a_0_2018;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);
/*Did Not Vote for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5 if year==2018 & d101d5a==0;
estimates store d101d5a_1_2018;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);
/*Would Vote for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5 if year==2018 & d101d5b==1;
estimates store d101d5b_0_2018;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all);
/*Would Not for Non-Party Candidate*/
#delimit;
set more off;
svy: reg  vote party gov_nominee age_39 male home VFF doctor lawyer bachelors masters doctorate trade poverty environment no_policy exp_zero exp_y5 if year==2018 & d101d5b==0;
estimates store d101d5b_1_2018;
outreg2 using "Tables\AppendixF", tdec(3) bdec(3) e(all) excel;

/************************GRAPH APPENDIX F*******************************/

#delimit;
coefplot d101d5a_0_2016 d101d5a_1_2016 d101d5b_0_2016 d101d5b_1_2016, keep(party)
xline(0, lcolor(red) lpattern(dash) lwidth(thick)) legend(rows(4) size(small) position(6) label(2 "Voted for Non-Party")
label(4 "Didn't Vote for Non-Party") label(6 "Would Vote for Non-Party") label(8 "Would Not Vote for Non-Party")) title("PAPI Survey 2016") scheme(s1mono)
msize(medlarge) mcolor(navy)
ylab("")
xlab(-.2(.05).4, labsize(small));
graph save "Figures\AppendixF_2016.gph", replace;

#delimit;
coefplot d101d5a_0_2017 d101d5a_1_2017 d101d5b_0_2017 d101d5b_1_2017, keep(party)
xline(0, lcolor(red) lpattern(dash) lwidth(thick)) legend(rows(4) size(small) position(6) label(2 "Voted for Non-Party")
label(4 "Didn't Vote for Non-Party") label(6 "Would Vote for Non-Party") label(8 "Would Not Vote for Non-Party")) title("PAPI Survey 2017") scheme(s1mono)
msize(medlarge) mcolor(navy)
ylab("")
xlab(-.2(.05).4, labsize(small));
graph save "Figures\AppendixF_2017.gph", replace;

#delimit;
coefplot d101d5a_0_2018 d101d5a_1_2018 d101d5b_0_2018 d101d5b_1_2018, keep(party)
xline(0, lcolor(red) lpattern(dash) lwidth(thick)) legend(rows(4) size(small) position(6) label(2 "Voted for Non-Party")
label(4 "Didn't Vote for Non-Party") label(6 "Would Vote for Non-Party") label(8 "Would Not Vote for Non-Party")) title("PAPI Survey 2018") scheme(s1mono)
msize(medlarge) mcolor(navy)
ylab("")
xlab(-.2(.05).4, labsize(small));
graph save "Figures\AppendixF_2018.gph", replace;

#delimit;
graph combine "Figures\AppendixF_2016.gph" "Figures\AppendixF_2017.gph" "Figures\AppendixF_2018.gph", cols(3) xcommon ycommon imargin(tiny) scheme(s1mono)
title("Marginal Probability of Vote for Party Member", size(medsmall) position(10) orientation(vertical))
note("Replicates Models 2, 6, 10 by Voter Type; Range bars=95% CIs.", size(vsmall));
graph save "Figures\AppendixF.gph", replace;


/****************************************All Versions of Economic Performance***********************************************/

#delimit;
set more off;
generate Bad=party if a011==0|a011==1;
generate Neutral=party if a011==2;
generate Good=party if a011==3|a011==4;


#delimit;
foreach x in Bad Neutral Good{;
svy: reg  vote `x' gov_nominee age_39 male home VFF trade poverty environment if year==2016;
*svy: prob  vote `x' gov_nominee age_39 male home VFF trade poverty environment if year==2016;
*mfx;
estimates store `x'16;
svy: reg  vote `x' gov_nominee age_39 male home VFF trade poverty environment no_policy if year==2017;
*svy: prob  vote `x' gov_nominee age_39 male home VFF trade poverty environment no_policy if year==2017;
*mfx;
estimates store `x'17;
svy: reg  vote `x' gov_nominee age_39 male home VFF trade poverty environment no_policy if year==2017;
*svy: prob  vote `x' gov_nominee age_39 male home VFF trade poverty environment no_policy if year==2017;
*mfx;
estimates store `x'18;
};

#delimit;
coefplot Good16 Neutral16 Bad16 , keep(Good Neutral Bad)
xline(0, lcolor(red) lpattern(dash) lwidth(thick)) legend(off) title("PAPI Survey 2016") scheme(s1mono)
msize(medlarge) mcolor(navy)
xlab(-.1(.1).3, labsize(small));
graph save "Figures\Figure4_2016.gph", replace;

#delimit;
coefplot Good17 Neutral17 Bad17 , keep(Good Neutral Bad)
xline(0, lcolor(red) lpattern(dash) lwidth(thick)) legend(off) title("PAPI Survey 2017") scheme(s1mono)
msize(medlarge) mcolor(navy)
xlab(-.1(.1).3, labsize(small));
graph save "Figures\Figure4_2017.gph", replace;

#delimit;
coefplot Good18 Neutral18 Bad18 , keep(Good Neutral Bad)
xline(0, lcolor(red) lpattern(dash) lwidth(thick)) legend(off) title("PAPI Survey 2018") scheme(s1mono)
msize(medlarge) mcolor(navy)
xlab(-.1(.1).3, labsize(small));
graph save "Figures\Figure4_2018.gph", replace;


#delimit;
graph combine "Figures\Figure4_2016.gph" "Figures\Figure4_2017.gph" "Figures\Figure4_2018.gph", cols(3) xcommon ycommon imargin(tiny) scheme(s1mono)
title("AMCE for Party Member", size(medsmall) position(9) orientation(vertical))
note("Based on Models 3, 7, and 11 in Table 5 at different levels of economic satisfaction; Range Bars=95% CIs.", size(small));
graph save "Figures\Figure4.gph", replace;
graph export "Figures\Figure4.pdf", replace;





/****************************************************************************************************************/
/*Appendix E with Histogram*/
#delimit;
histogram a011 if a011<6 & year==2016, discrete percent fcolor(navy) lcolor(white) 
xtitle(Present Economic Situation of Respondent) xtitle("", size(medium) margin(medium)) 
xlabel(, labels labsize(small) angle(forty_five) valuelabel) xlab(0(1)4) note("")
ytitle("Percentage of Respondents", size(medium) margin(medium)) title("2016")
note("") scheme(s1mono);
graph save "Figures\AppendixE_2016.gph", replace;

#delimit;
histogram a011 if a011<6 & year==2017, discrete percent fcolor(navy) lcolor(white) 
xtitle(Present Economic Situation of Respondent) xtitle("", size(medium) margin(medium)) 
xlabel(, labels labsize(small) angle(forty_five) valuelabel) xlab(0(1)4) note("")
ytitle("Percentage of Respondents", size(medium) margin(medium)) title("2017")
note("") scheme(s1mono);
graph save "Figures\AppendixE_2017.gph", replace;

#delimit;
histogram a011 if a011<6 & year==2018, discrete percent fcolor(navy) lcolor(white) 
xtitle(Present Economic Situation of Respondent) xtitle("", size(medium) margin(medium)) 
xlabel(, labels labsize(small) angle(forty_five) valuelabel) xlab(0(1)4) note("")
ytitle("Percentage of Respondents", size(medium) margin(medium)) title("2018")
note("") scheme(s1mono);
graph save "Figures\AppendixE_2018.gph", replace;

#delimit;
graph combine "Figures\AppendixE_2016.gph" "Figures\AppendixE_2017.gph" "Figures\AppendixE_2018.gph", xcommon ycommon imargin(tiny) cols(3)  scheme(s1mono)
title("Present Economic Situtation of Respondent", position(6) size(medium));
graph save "Figures\AppendixE.gph", replace;















